Язык программирования Тривиль

Версия языка: 0.9.1

Алексей Недоря 25.09.2023

Содержание

1	Назн	пачение	5
2	Обзо	р языка	6
3	Леко 3.1 3.2 3.3 3.4 3.5 3.6 3.7 3.8 3.9 3.10	Комментарии	7 7 8 8 8 9 9 9 10
4	4.1 4.2 4.3 4.4 4.5 4.6 4.7 4.8 4.9	4.11.1 Параметры	11 11 12 12 13 13 14 15 16 16 17 17 17 18 18 19 20 20 20
5		ажения Первичные выражения Операнды Доступ к полям и методам Индексация Преобразование типа Вызов функции или метода 5.6.1 Развернутое выражение Конструктор вектора	22 22 22 23 23 24 25 26 26 28

	5.9	Подтверждение типа
	5.10	Унарные операции
		Проверка типа
		Бинарные операции
	5.12	5.12.1 Приоритеты операции
		5.12.1 Приоритель операции 31 5.12.2 Арифметические операции 31
		5.12.2 Арифметические операции 31 5.12.3 Операции сравнения 31
		5.12.3 Операции сравнения
		1 ,
		± '
	c 12	5.12.6 Операции сдвига
	5.13	Константные выражения
6	Опер	раторы 32
	6.1	Блоки
	6.2	Локальные описания
	6.3	Выражение, как оператор
	6.4	Оператор присваивания
	6.5	Инкремент и декремент
	6.6	Оператор 'если'
	6.7	Оператор 'надо'
	6.8	Оператор 'выбор'
	0.6	1 1 1
		1 1
		1 1 1
	<i>(</i> 0	1 2
	6.9	Оператор 'пока'
		Оператор цикл
		Оператор 'прервать'
		Оператор 'вернуть'
	6.13	Оператор 'авария' и аварийная ситуация
7	Стан	ідартные функции и методы 41
	7.1	Стандартные функции
		7.1.1 Длина
		7.1.2 Функции для полиморфных параметров
	7.2	Встроенные методы для векторов
		7.2.1 Метод добавить
8	Мод	ули 43
	8.1	Заголовок модуля
	8.2	Импорт
	8.3	Вход или инициализация модуля
	8.4	Инициализация и исполнение программы
9	Ofor	бщенные модули 46
,	9.1	Последствия принятого решения
	9.1	
	9.2	Уточненный синтаксис модуля
10		вила совместимости 49
	10.1	Эквивалентность типов
	10.2	Совместимость по присваиванию

Применять с осторожностью	50
11.1 Осторожные преобразования типа	50
Версии языка	51
А.1 Версия 0.9.0 от 07.09.2023	51
А.2 Версия 0.9.1 от 07.09.2023	51

1. Назначение

Язык программирования Тривиль разработан в рамках работы на семейством языков программирования Языки выходного дня (ЯВД) проекта Интенсивное программирования.

Тривиль является предварительным языком семейства ЯВД, предназначенным для реализации компиляторов и экосистемы следующих языков семейства. В рамках классификации языков, принятом в проекте Интенсивное программирования, это язык L2.

Основными требованиями к языку были поставлены:

- язык должен быть минимально достаточным, то есть в него должны быть включены только те типы и конструкции, которые необходимы для реализации компиляторов
- язык должен быть современным с точки зрения набора конструкций
- язык должен поддерживать надежное программирование (автоматическое управление памятью, отсутствие неопределенного поведение, безопасность указателей, минимизация неявных конструкций)
- язык должен обеспечивать легкость чтения и понимания (readability) и легкость разработки
- как следствие: язык должен быть русскоязычным. Лексика и синтаксис языка должны минимизировать переключение на латиницу и обратно в процессе разработки программ

Название языка происходит от слова "тривиальный что означает, что при разработке языка практически везде использовались решения, проверенные в других современных языках программирования, в первую очередь "донорами"являются Go, Swift, Kotlin и Oberon.

В итоге разработки, несмотря на первоочередную направленность на разработку компиляторов, Тривиль является языком программирования общего назначения, пригодным для решения широкого круга задач.

Получившийся язык (и экосистема) обладает существенными достоинствами для использования в качестве полигона для обучения студентов разработке компиляторов, библиотек, алгоритмов оптимизации и так далее, в первую очередь это:

- Простота языка
- Современный вид и набор конструкций языка
- Простота компилятора
- Открытая лицензия

2. Обзор языка

Тривиль - это модульный язык с явным экспортом и импортом, с поддержкой ООП и автоматическим управлением памятью (сборка мусора).

Программа на языке Тривиль состоит из модулей (единиц компиляции), исходный текст каждого модуля расположен в одном или нескольких исходных файлов.

Пример программы:

```
1 модуль х

2 импорт "стд/вывод"

4 вход {

 вывод.ф("Привет!\n")

7 }
```

Для описания языка используется EBNF в формате, близком к формату ANTLR4. Операции:

```
() группировка

X* повторение 0 и более раз

X+ повторение 1 и более раз

X? опциональность X (0 или 1 раз)
```

Х | Ү | Х или Ү

Пример:

Список-операторов: Оператор (Разделитель Оператор)*

3. Лексика

Исходный текст есть последовательность лексем: идентификаторов (§3.3), ключевых слов (§3.4), знаков операций и знаков препинания (§3.5), литералов (§3.6, §3.7, §3.8, §3.9) и модификаторов (§3.10). Каждая лексема состоит из последовательности Unicode символов (unicode code point) в кодировке UTF-8.

Пробелы (U+0020), символы табуляции (U+0009) и символы завершения строки (U+000D, U+000A) разделяют лексемы, и, игнорируются, кроме следующих случаев:

- Символы завершения строк могут использоваться как разделители синтаксических конструкций (§3.2).
- Пробелы являются значащими символами в идентификаторах, состоящих из нескольких слов (§3.3).
- Пробелы являются значащими в строковых и символьных литералах (§3.3).

Несколько разделителей трактуются, как один.

Исходный текст может содержать комментарии (§3.1).

3.1. Комментарии

Есть две формы комментариев:

- Строчный комментарий начинается с последовательности символов '//' и заканчивается в конце строки.
- Блочный комментарий начинается с последовательности символов '/*' и заканчивается последовательностью символов '*/'. Блочные комментарии могут быть вложенные.

```
Комментарий
: '//' (любой символ, кроме завершения строки)*
| '/*' (любой символ)* '*/'
```

3.2. Разделители синтаксических конструкций

Некоторые синтаксические правила используют нетерминал *Разделитель* для разделения двух подряд идущих синтаксических конструкций, например:

```
Список-операторов: Оператор (Разделитель Оператор)*
```

В качестве разделителя может использоваться символ ';' или символ завершения строки.

```
Разделитель: ';' | символ-завершения-строки
```

Пример:

```
1 a := 1; 6 := 2
2 B := 1
```

В строке 1 операторы разделены символом ';', а оператор в строке 2 отделен от операторов строки 1 символом завершения строки.

Ошибка компиляции - нет разделителя:

```
a := 1 f := 2
```

3.3. Идентификаторы

Идентификатор - это последовательность *Слов*, разделенных пробелами или символами дефис '-' с опционально завершающим знаком препинания:

Каждое слово состоит из *Букв* и $\mu \mu p$, и начинается с Буквы. Буквой считается любой Unicode символ, имеющий признак *Letter*, и, дополнительно, символы ' ν ' и ' '.

```
Идентификатор: Слово ((' ' | '-') Слово)* Знак-препинания? Слово: Буква (Буква | Цифра)* Буква: Unicode-letter | '_' | 'N' ' Цифра: '0' .. '9' Знак-препинания: '?' | '!'
```

Примеры идентификаторов:

```
1 буква
2 буква-или-цифра
3 №-символа
4 Цифра?
5 Пора паниковать!
```

3.4. Ключевые слова

Следующие ключевые слова зарезервированы и не могут быть использованы, как идентификаторы:

авария	если	конст	пока	ТИП
вернуть	иначе	мб	прервать	типа
вход	импорт	модуль	пусть	фн
выбор	класс	надо	позже	цикл
другое	когда	осторожно	среди	

3.5. Знаки операций и знаки препинания

Следующие последовательности символов обозначают знаки операций и знаки препинания:

```
+ - * / %
= # < <= > >=
```

```
:& :| :\ :~ << >>
:= ++ --
( ) [ ] { }
(: . ^ , : ;
```

3.6. Целочисленные литералы

Тип десятичного литерала: Цел 64, тип шестнадцатеричного литерала: Слово 64 (§4.3).

3.7. Вещественные литералы

В текущей реализации есть только одна форма записи вещественных литералов, без экспоненты.

```
Вещественный-литерал: Цифра+ '.' Цифра*
```

Тип вещественного литерала: Вещ64 (§4.3).

3.8. Строковые литералы

Строковый литерал - это последовательность символов, заключенные в двойные кавычки. Строковый литерал может содержать символы, закодированные с помощью еscape-последовательности, которая начинается с символа '\'.

```
Строковый-литерал

: '""
    (~('"" | '\\" | '\n" | '\r" | '\t') | Escape)*

Escape

: '\\"
    ( 'u' Цифра16 Цифра16 Цифра16 Цифра16 | 'n' | 'r" | 't' | '""
    | """
    | """
    | """
    | """
```

Тип строкового литерала: Строка (§4.3).

3.9. Символьные литералы

Символьный литерал задает значение для Unicode code point, - это последовательность символов, заключенные в двойные кавычки. Он записывается как один или несколько символов, заключенных в одинарные кавычки. Символьный литерал может быть закодирован с помощью escape-последовательности, которая начинается с символа '\'.

Тип символьного литерала: Символ (§4.3).

3.10. Модификаторы

Модификаторы использутся в исходном тексте, чтобы внести изменение в в семантику и/или синтаксис конструкции языка, см., например, §4.11.3.

```
Модификатор: '@' Буква+ Список-атрибутов?
Список-атрибутов: '(' (Атрибут (',' Атрибут)*)? ')'
Атрибут: Строковый-литерал ': 'Строковый-литерал
```

```
@внеш("имя":"print_string")
```

4. Описания и области действия

Каждый идентификатор, встречающийся в программе, должен быть описан, если только это не предопределенный идентификатор (§4.2). Идентификатор может быть описан как тип, константа, переменная, функция или метод, а так же как параметр в сигнатуре функции или метода, и как поле в типе класса.

Описание

: Описание-типов

| Описание-констант

| Описание-переменных

| Описание-функций

| Описание-метода

4.1. Области действия

Описанный идентификатор используется для ссылки на связанный объект, в тех частях программы, которые попадают в *область действия* описания. Идентификатор не может обозначать более одного объекта в пределах заданной области действия. Область действия может содержать внутри себя другие области действия, в которых идентификатор может быть переопределен.

Область действия, которая содержит в себе все исходные тексты на языке Тривиль называется *Универсум*.

Области действия:

- Областью действия предопределенного идентификатора является Универсум
- Областью действия идентификатора, описанного на верхнем уровне (вне какой-либо функции), является весь модуль (§8).
- Областью действия идентификатора импортируемого модуля является файл (часть модуля), содержащего импорт (§8.2).
- Областью действия идентификатора, обозначающего параметр функции, является тело функции (§4.11).
- Областью действия идентификатора, описанного в теле функции (§4.11) или теле входа (§8.3), является часть блока (§6.1), в котором описан идентификатор, от точки завершения описания и до завершения этого блока.

В описании верхнего уровня за идентификатором может следовать признак экспорта '*', указывающий, что идентификатор экспортирован и может использоваться в другом модуле, *импортирующем* данный (§8.2).

Правило для опционально экспортированного идентификатора:

Идент-оп: Идентификатор '*'?

4.2. Предопределенные идентификаторы

Следующие идентификаторы неявно описаны в области действия Универсум.

```
Типы (§4.3):

Байт Цел64 Слово64 Вещ64 Лог Символ Строка Строка8 Пусто Константы типа Лог (§4.3):

ложь истина

Литерал, обозначающий отсутствие значения для может быть типа (§4.10):

пусто

Стандартные функции (§7.1):

длина тег нечто

Кроме того, для векторных типов определен набор встроенных методов (§7.2).
```

4.3. Предопределенные типы

Следующие типы обозначаются предопределенными идентификаторами, значениями данных типов являются:

Тип	Множество значений		
Байт	множество целых чисел от 0 до 255		
Цел64	множество всех 64-битных знаковых целых чисел		
Слово64	множество всех 64-битных беззнаковых целых чисел		
Вещ64	множество всех 64-разрядных чисел с плавающей запятой стандарта IEEE-754		
Лог	константы ложь и истина		
Символ	множество всех Unicode символов		
Строка	множество всех строковых значений (§4.3.1)		
Строка8	множество всех строковых значений (§4.3.1)		
Пусто	литерал пусто		

Операции над значениями этих типов определены в (§5.10, §5.12).

4.3.1. Тип Строка

Строковый тип представляет собой набор строковых значений. Строковое значение - это (возможно, пустая) последовательность Unicode символов в кодировке UTF-8. Количество символов называется длиной строки и никогда не бывает отрицательным. Строки неизменяемы: после создания изменить содержимое строки невозможно. Количество символов строки можно определить с помощью встроенной функции длина (§7.1.1).

Операция индексации для строк не определена:

```
1 пусть с = "привет"
2 пусть сим = c[0] // ошибка
```

Для работы с байтами строки можно использовать тип Строка8 (§4.3.2).

4.3.2. Тип Строка8

Тип Строка8 введен для удобной и быстрой работы со строками в стандартных библиотеках. Единственным способом получения значения типа Строка8 является преобразование из строки. Это преобразование выполняется только во время компиляции.

```
пусть с8 = "Привет"(:Строка8)
```

Значение типа Строка8 - это неизменяемая, индексируемая последовательность байтов. Стандартная функция длина примененная к значению типа Строка8 возвращает число байтов.

Сравнение типов:

	Строка	Строка8
Определение	последовательность	последовательность
	символов	байтов
Индексация	нет	да
длина(с)	число символов	число байтов
можно менять	нет	нет

```
1 пусть с = "Привет"
2 пусть с8 = с(:Строка8)
3 
4 вывод.ф("%v\n", длина(с)) // выведет: 6
5 вывод.ф("%v\n", длина(с8)) // выведет: 12
6 вывод.ф("%v\n", с8[0]) // выведет: 0xD0
```

Значение типа Строка8 неизменяемое, менять байты в нем нельзя:

```
1 пусть с8 = "Привет" (:Строка8)
2 с8[0] := 1 // ошибка
```

4.4. Указание типа

Тривиль является языком со статической типизацией, что означает, что тип любого объекта языка явно или неявно указывается во время описания объекта. Неявное указание типа может быть использовано в описании констант ($\S4.5$), переменных ($\S4.5$) и полей класса ($\S4.9$).

Для явного указании типа используется имя типа, перед которым может стоять ключевое слово мб (*может быть*).

```
Указ-типа: 'мб'? Имя-типа
Имя-типа: Идентификатор | Импорт-идент
Импорт-идент: Идентификатор '.' Идентификатор
```

Если в качестве имени типа используется *Импорти-идент*, то первый идентификатор в нем должен обозначать имя импортированного модуля (§8.2), а второй идентификатор должен быть идентификатором экспортированного из этого модуля объекта, в данном контексте - идентификатор типа.

Множество значений объекта с типом мб T состоит из значения, обозначенного предопределенным идентификатором пусто и значений типа T. Тип такого объекта называется может быть T (§4.10).

4.5. Описание констант

Описание константы связывает идентификатор с постоянным значением. Значение константы может быть задано явно или неявно, в случае группового описания констант.

```
Описание-констант: 'конст' (Константа | Группа-констант)
Константа: Идент-оп (':' Указ-типа)? '=' Выражение
```

Если тип константы не указан, то он устанавливается равным типу выражения (§5). Выражение для константы должно быть константным выражением (§5.13).

```
1 конст к1: Цел64 = 1 // тип Цел64
2 конст к2: Байт = 2 // тип Байт
3 конст к3 = 3 // тип Цел64
4 конст к4 = "Привет" // тип Строка
```

Групповое описание констант позволяет опускать тип и выражение для всех констант, кроме первой константы в группе и указать признак экспорт для всех констант группы.

```
Группа-констант:
    '*'? '('
    Kонстанта (Разделитель След-константа)*
    ')'
След-константа: Идент-оп ((':' Указ-типа)? '=' Выражение)?
```

Пример группового экспорта:

```
1 конст *(
2 Счетчик = 1
3 Имя = "Вася"
4 )
```

Пример неявного задания значения для констант:

```
1 конст (
2 / / операции
3 ПЛЮС = 1 // тип Цел64, значение = 1
4 МИНУС // тип Цел64, значение = 2
5 ОСТАТОК // тип Цел64, значение = 3
```

4.6. Описание переменных

Описание переменной создает переменную, привязывает к ней идентификатор и указывает её тип и начальное значение через *Инициализацию*.

```
Описание-переменной:
    'пусть' Идент-оп (':' Указ-типа)? Инициализация
Инициализация: (':=' | '=') ('позже' | Выражение)
```

Если тип переменной не указан, то он устанавливается равным типу выражения (§5).

Каждая переменная должна быть явно проинициализирована. Если инициализация задана через лексему '=', то значение переменной не может быть изменено (переменная с единственным присваиванием), если же в инициализации используется лексема ':=', то значение может быть изменено (§6.4, §6.5), а переменная называется изменяемой.

Явное указание типа:

```
1 пусть ц1: Цел64 = 1
2 пусть ц2: Цел64 := 2
```

Неявное указание типа:

```
1 пусть ц3 = 3 // неявное указание типа
2 пусть ц4 := 4
3 пусть имя = Имя языка()
```

Компилятор выдает ошибку, при попытке изменить значение переменной с единственным присваиванием:

```
1 ц1 := 2
2 ц3++
3 имя := "C++"
```

Если в инициализации задано *Выражение*, то начальным значение переменной является значение выражения. Если при этом тип переменной явно задан, то выражение должно быть *совместимо по присваиванию* с типом переменной (§10.2).

Вместо выражения может быть указано ключевое слово позже, это *поздняя инициализация*. Такая форма инициализации разрешена только для переменных уровня модуля, при этом тип таких переменных, должен быть явно указан.

Значение переменной с поздней инициализацией должно быть задано в инициализации модуля (§8.3). Текущая реализация компилятора запрещает позднюю инициализацию для переменных с единственным присваиванием.

4.7. Описание типов

Описание типа связывает идентификатор с новым или существующим типом.

```
Описание-типа:
    'тип' Идент '='
    ( Тип-вектора
    | Тип-класса
    | Указ-типа
)
```

Если в описание типа указан тип вектора (§4.8) или тип класса (§4.9), то создается новый тип и определяется структура данных этого типа, и, как следствие, набор операций над данными этого типа.

Если в описании типа стоит указание на тип ($\S4.4$), то это определение еще одного имени для уже существующего типа или задание имени для *типа вектора* ($\S4.8$) или *может быть типа* ($\S4.10$).

```
1 тип Цел = Цел64
2 тип Текст= []Строка
3 тип Строка? = мб Строка
```

Так как указание типа не создает новый тип, использование типа Цел из примера идентично использованию типа Цел 64.

4.8. Тип вектора

2

Вектор - это пронумерованная последовательность элементов одного типа, называемая типом элемента. Количество элементов называется длиной вектора, длина не может быть отрицательной. Элементы вектора доступны через операцию индексации (§5.4). Для индексации вектора используются целочисленные индексы от 0 до длина-1. Векторы всегда одномерны, но типом элемента может быть вектор, там самым формируя многомерную структуру.

```
Тип-вектора: '[' ']' Указ-типа

тип Байты = []Байт

тип Матрица = [][]Вещ64
```

Длина вектора может изменятся во время выполнения, другими словами, вектор - это *динамический массив*. Для получения текущей длины вектора используется стандартная функция длина (§7.1.1). Начальное значение для объекта типа вектор задается с помощью конструктора вектора (§5.7). Далее, длину вектора можно поменять с помощью стандартных методов (§7.2).

Пример инициализации вектора из трех элементов:

```
1 пусть байты = Байты[1, 2, 3]
```

4.9. Тип класса

Тип класса - это структура, состоящая из полей. С типом класса могут быть связаны функции, называемые методами (§4.12).

Описание класса состоит из опционального указания базового класса и списка полей. Если базовый класс указан, то класс *наследует* поля и методы базового класса (§4.9.3).

Для каждого поля в списке задается идентификатор, тип (явно или неявно) и начальное значение. Областью действия идентификаторов полей является само описание класса, но они могут быть доступны в операции доступа к полям и методам (§5.3).

```
Тип-класса: 'класс' Базовый-класс? '{' Список-полей? '}' Базовый-класс: '(' Указ-типа ')' Список-полей: Поле (Разделитель Поле)*
Поле: Идент-оп (':' Указ-типа)? Инициализация
```

```
тип Человек = класс {
2 имя: Строка := ""
3 возраст: Цел64 := 0
4 }
```

Если тип поля не указан, то он устанавливается равным типу выражения (§5).

4.9.1. Инициализация полей

Каждое поле класса должно быть явно проинициализировано. Если инициализация задана через лексему '=', то значение поля не может быть изменено (*none с единственным присваиванием*), если же в инициализации используется лексема ':=', то значение поля может быть изменено ($\S6.4$, $\S6.5$), а поле называется *изменяемым*.

Если в инициализации задано *Выражение*, то начальным значение поля является значение выражения. Если при этом тип поля явно задан, то выражение должно быть *совместимо по присваиванию* с типом поля (§10.2).

Вместо выражения может быть указано ключевое слово позже, это *поздняя инициализация*. Такая форма инициализации разрешена только для полей, тип которых явно указан.

Значение поля с поздней инициализацией должно быть задано при создании экземпляра класса в конструкторе экземпляра класса (§5.8).

4.9.2. Экспорт полей

Если тип класса экспортируется, его поля могут быть помечены признаком экспорта, такие поля называется экспортированными полями. Поля, которые не экспортированы, доступны только в том модуле, в котором описан тип класса.

4.9.3. Наследование

Наследование позволяет определить новый (*расширенный*) класс на основе существующего (базового) класса.

```
1 тип Работник = класс (Человек) {
2 зарплата := 0.0
3 }
```

Базовый класс *Человек*, указанный в описании класса *Работник*, называется *прямым базовым классом*. Так как *Человек* может быть, в свою очередь, расширением другого базового класса, для каждого класса определен список базовых классов, возможно, пустой. Термин *базовый класс* будет использовать для обозначения любого класса из этого списка. Циклы в списке базовых классов запрещены.

Расширенный класс наследует поля и методы базового класса и может добавлять новые поля и методы. Обращение к полям и методам базового класса ничем не отличается от обращения к полям и методам самого класса. Методы базового класса можно переопределять, сохраняя те же самые типы параметра и тип результата (Seecompat:equal-types).

4.10. Может быть тип

Как правило, современные языки программирования ограничивают работу со объектами ссылочных (reference) типов для того, чтобы сделать явными все места в программе, в которых может возникнуть ошибка использования нулевой ссылки (null pointer exception). Так как русская терминология в этой области не устоялась, приходится обращаться к англоязычным терминам.

Язык Тривиль следует уже выработанному в современных языках программированию подходу (но не синтаксису):

- Иначе, если ключевое слово мб отсутствует в указании типа объекта, значением этого объекта может быть только значений типа T.

Ссылочными типами являются Строка, типы вектора и типы класса, к остальным типам мб не может применятся.

Для объекта типа $m \delta T$ определены следующие действия:

- Присваивание объекту значения 'пусто' или значения типа Т (§10.2)
- Сравнение на равно/не равно с 'пусто' или с другим объектом типа $m \delta T$ (§5.12.3)
- Операция подтверждения типа ' ^ ' (§5.9), позволяющая перейти от значения типа $m\delta$ T к значению типа T

Пример указания типа и использования объекта:

```
1 пусть кличка: мб Строка := пусто
2 ...
3 если кличка # пусто { вывод.ф("%v\n", кличка^}
```

4.11. Описание функций

Описание функции состоит из идентификатора, сигнатуры и тела функции. Сигнатура определяет формальные параметры и тип результата (если таковой имеется). Особым видом функции является метод, см. §4.12.

```
Описание-функции: 'фн' Идент-оп Сигнатура Тело
Сигнатура: '(' Список-параметров? ')' Тип-результата?
Тип-результата: ':' ('*' | Указ-типа)
Тело: (Блок | Модификатор)?
```

Вместо тела функции может стоять модификтор (§3.10) @внеш, см. §4.11.3. Если у функции задан тип результата, тело функции должно завершаться операторами вернуть (§6.12) или авария (§6.13).

Если в типе результата вместо указания типа использован символ '*', то функция возвращает полиморфный результат. Такая функция может возвращать значение любого типа. Для работы с полиморфным значением определены специальные стандартные функции (§7.1.2).

```
1 фн Факториал(ц: Цел64): Цел64 {
2 если ц <= 1 { вернуть 1 }
3 вернуть ц * Факториал(ц - 1)
4 }
```

4.11.1. Параметры

Формальные параметры - это идентификаторы, которые обозначают *аргументы* (фактические параметры), указанные при вызове функции (§5.6).

```
Список-параметров: Параметр (',' Параметр)* ','?
Параметр:
Идентификатор (':' | ':=') '...'? ('*' | Указ-типа)
```

Параметры бывают двух видов: *входные* и *выходные* параметры. Вид параметра задается символом после идентификатора, задающего имя параметра. Для входных параметров это символ ':-', для выходных - символ ':--'. Выходной параметр, в более привычной терминологии, является *in out* параметром.

Аргументом для входного параметра должно быть выражение, тип которого совместим по присваиванию с типом параметра. Аргументом для выходного параметра должно быть *изменяемое выражение* ($\S 6.4$), тип которого эквивалентен типу параметра.

В теле функции, входной параметр имеет значение, которое передается в функцию, изменяется функцией и передается обратно из функции для замены исходного значения.

```
фн извлечь число(стр: Строка, число:= Цел64): Лог {...}

пусть рез := 0

если ~ извлечь число(с, рез) {

вывод.ф("не удалось извлечь число")

}
```

Последний параметр функции может иметь тип с префиксом . . . - это вариативный параметр. Функция с таким параметром может быть вызвана с нулем или более аргументов для этого параметра. Если тип вариативного параметра указан как T, то в теле функции типом параметра является [] Т.

```
фн сумма(числа: ...Цел64): Цел64 {...}

// В вызове можно указать любое число аргументов

ответ := сумма() // без аргументов

ответ := сумма(1, 2, 3) // три аргумента
```

4.11.2. Полиморфные параметры

Если вместо типа параметра указан символ '*', то параметр называется полиморфным. Аргумент, соотвествующий этому параметру, может быть выражением любого типа. Для вариативного полиморфного параметра каждый аргумент может быть выражением любого типа. Для работы с полиморфным параметром определены специальные стандартные функции (§7.1.2).

```
ф ф печать по формату (формат: Строка, аргументы: ...*) {
    ...
    }
```

4.11.3. Внешние функции

Функция, в которой вместо тела стоит модификатор @внеш, является внешней функцией, то есть реализованной каким-то внешним способом. Атрибут "имя" модификатора задает внешнее имя функции.

```
фн строка(c: Строка) @внеш("имя":"print_string")
```

Если атрибут "имя" не задан, внешнее имя совпадает с идентификатором функции.

4.12. Описание методов

Метод - это функция, связанная с типом классом. Для вызова метода (§5.6) должен быть указан экземпляр этого класса или расширенного класса. В описание метода класс, с которым связан метод указывается с помощью Π ривязки.

```
Описание-метода: 'фн' Привязка Идент-оп Сигнатура Блок Привязка: '(' Идентификатор ':' Указ-типа ')'
```

Привязка определяет идентификатор и тип, который должен быть типом класса. В теле функции идентификатор привязки является параметром указанного типа.

```
1 тип K = класс {}
2 фн (к: K) метод() {}
```

Идентификатор метода должен быть уникальным в классе среди идентификатором полей и методов.

В расширенном классе может быть определен метод с таким же идентификатором как в одном из базовых классов.

переопределен, то есть опеределен другой метод . При этом сигнатура переопределеного метода должна совпадать с сигнатуром переопределяемого метода, а именно:

- Число параметров должно совпадать
- Типы результата должны быть равны §10.1
- Типы параметров должны быть равны §10.1
- Признаки вариативности и полиморфности для каждого параметра должны совпадать

Пример ошибки при переопределении метода (разное число параметров):

```
1 тип K1 = класс (K) {}
2
3 фн (к: K1) метод(с: Строка) {}
```

5. Выражения

Выражение определяет вычисление значения путем применения операций к операндам.

```
Выражение
: Унарное-выражение
| Проверка-типа
| Выражение Бинарная-операция Выражение

Унарное-выражение
: Первичное-выражение
| Унарная-операция Унарное-выражение

Проверка-типа: Выражение 'типа' Указ-типа
```

5.1. Первичные выражения

Первичные выражения - это выражения, которые являются операндами унарных ($\S 5.10$) и бинарных ($\S 5.12$) операций .

```
Первичное-выражение

: Операнд
( Доступ
| Индексация
| Преобразование
| Вызов
| Конструктор-вектора
| Конструктор-класса
| Подтверждение-типа
) *
```

5.2. Операнды

Операнды - это элементарные значения в выражении. Операнд может быть литералом, именем объекта, обозначающим константу, переменную или функцию, или выражением в скобках.

```
Операнд: Литерал | Имя-объекта | '(' Выражение ')'
Литерал
: Целочисленный-литерал
| Вещественный-литерал
| Строковый-литерал
| Символьный-литерал
Имя-объекта: Идентификатор | Импорт-идент
Импорт-идент: Идентификатор '.' Идентификатор
```

Если в качестве имени объекта используется *Импортидент*, то первый идентификатор в нем должен обозначать имя импортированного модуля (§8.2), а второй идентификатор должен быть идентификатором экспортированного из этого модуля объекта.

Операнд	Тип выражения
Имя объекта	тип объекта
Целочисленный-литерал	Цел64
Вещественный-литерал	Вещ64
Строковый-литерал	Строка
Символьный-литерал	Символ
Выражение в скобках	тип выражения

5.3. Доступ к полям и методам

Доступ применяется к первичному выражению.

```
Доступ: '.' Идентификатор
```

В выражении что-то.имя:

- что-то должно быть выражением типа T, где T это тип класса.
- а имя должно быть идентификатором поля или метода типа T или базового класса T

Типом выражения доступа является тип поля или метода.

```
1 чел.возраст := 5
2 если чел.возраст < 18 { ... }
```

5.4. Индексация

Индексация применяется к первичному выражению.

```
Индексация: '[' Выражение '['
```

Выражение что-то [индекс] обозначает элемент вектора или вариативного параметра, индекс которого определяется выражением индекс.

- что-то должно быть идентификатором, обозначающим вариативный параметр
- или что-то должно быть выражением типа T, где T это тип вектора
- индекс должен быть выражением типа Цел 64 или Байт

Если во время исполнения индекс выходит за границы вектора или параметра (индекс $< 0 \mid$ индекс >= длина), происходит *авария* (§6.13).

Типом выражения индексации является тип элемента вектора, если индексируется вектор или тип вариативного параметра, если индексируется вариативный параметр.

Так как индексация и конструктор вектора ($\S5.7$) не всегда различимы синтаксически, выбор между ними происходит на уровне семантики, если выражение что-то обозначает тип, то это конструктор вектора, иначе индексация.

5.5. Преобразование типа

Преобразование типа применяется к первичному выражению, и преобразует, если это возможно, значение выражения к *целевому типу*, указанному в операции преобразования.

```
Преобразование: '(:' 'осторожно'? Указ-типа ')'
```

Если в преобразование есть ключевое слово осторожно, то это небезопасная операция, которая рассматривается отдельно (§11.1).

Часть преобразований, например, преобразования между числовыми типами и символами, а так же строковые преобразования могут изменить представление значения первичного выражения и повлечь за собой затраты на выполнение, а так же *аварийное завершение*, если выполнить преобразование невозможно.

Преобразование классов изменяют только тип, но не представление значения.

Разрешенные преобразования:

Целевой тип	Тип выражения	Условие выполнения
Байт	Цел64, Слово64, Символ,	значение в диапазоне
		0255
	Байт, Символ, Строковый ли-	всегда
Цел64	терал длины 1	
	Слово64	значение в диапазоне
		0Мах(Цел64)
	Вещ64	всегда
Слово64	Байт, Символ, Строковый ли-	всегда
Словооч	терал длины 1	
	Цел64	не отрицательное значение
Вещ64	Цел64	всегда
Символ	Байт, Строковый литерал дли-	всегда
Символ	ны 1	
	Цел64, Слово64	значение в диапазоне
		Unicode символа
Строка	Символ, [] Символ, [] Байт	всегда
[] Байт	Строка, Символ	всегда
[] Символ	Строка	всегда
Класс Ц	Класс К	К является базовым клас-
Класс Ц		сом класса Ц
	мб К, где К - класс	значение не равно пусто, и
		К является базовым клас-
		сом класса Ц

```
1 тип Байты = []Байт
2 пусть байты = "Привет"(:Байты)
3 пусть три = pi(:Цел64)
```

5.6. Вызов функции или метода

Вызов применяется к первичному выражению. Это выражение

- является идентификатором стандартной функции (§7.1)
- или обозначает функцию, тогда это вызов функции
- или явлется выражением типа класс, тогда это вызов метода

```
Вызов: '(' Список-аргументов? ')'
Список-аргументов: Аргумент (',' Аргумент)* ','?
Аргумент: Выражение '...'?
```

В вызове указывается список аргументов. Число аргументов в списке должно быть равно числу параметров в сигнатуре функции или метода, за исключение случая, когда в сигнатуре есть вариативный параметр. Для вариативного параметра число аргументов может быть от нуля и более.

Каждый аргумент в вызове является выражением. Если для вариативного параметра задан единственный аргумент, то после выражения может стоять знак операции . . . , делающий развернутое выражение (§5.6.1).

Для всех других аргументов, выражение должно быть *совместимо по присва- иванию* с типом параметра (§10.2).

Типом выражения вызова является тип результата. Если в сигнатуре функции или метода тип результата отсутствует, то вызов не может быть операндом для любой операции.

```
1 фн Факториал(ц: Цел64): Цел64 { /*тело*/ }
2 
3 пусть рез = Факториал(5)
```

5.6.1. Развернутое выражение

Развернутое выражение может быть использовано только как аргумент для вариативного параметра и только, если это единственный аргумент для этого параметра. Если развернутое выражение $x\dots$ используется как аргумент для параметра $\pi:\dots$ Т, то должно выполняется одно из условий:

- x является вариативным параметром того же типа $\mathbb T$
- x это выражение типа вектора, причем тип вектора определен, как [] Т

Пример использования развернутого вектора в качестве аргумента вариативного параметра:

```
фн Соединить (список: ...Строка) {/*тело*/}

тип Строки = []Строка

фн Сохранить (строки: Строки) {
пусть текст = Соединить (строки...)
/* сохранение текста */

}
```

5.7. Конструктор вектора

Конструктор вектора применяется к первичному выражению, которое должно обозначать тип вектора. Результатом выполнения конструктора является вектор.

Простой конструктор вектора задает последовательность выражений, каждое из которых должно быть совместимым по присваиванию с типом элемента вектора ($\S10.2$). Первое выражение в последовательности определяет значение элемента вектора с индексом 0, и так далее. Если ни одного значения не задано, то длина вектора равно нулю.

```
тип Числа = []Цел64

пусть ч = Число[] // вектор длины 0
пусть ч1 = Число[1, 2, 3] // вектор длины 3
```

Пример использования конструктора вектора в вызове:

```
тип Строки = []Строка
ф Сохранить (строки: Строки) { /*тело*/ }

Сохранить (Строки["привет", "мир"])
```

Конструктор вектора может быть задан последовательностью Элементов, где каждый элемент - это пара, разделенная двоеточием. Если хотя бы один элемент задан парой, то все элементы конструктора должны быть также заданы парами.

Для *Элемента* X: выражение:

X	выражение
длина задает длину конструируемого вектора	
выделить	задает число элементов, выделяемых для конструируемо-
	го вектора
*	задает значение по умолчанию для всех элементов, для
	которых оно не задано явно
Индекс	задает значение элемента с индексом Индекс

Следующие условия должны выполняться для вектора, заданного *Элемента-ми*:

- выражение для длина должно быть *совместимым по присваиванию* с типом Цел64
- выражение для выделить должно быть совместимым по присваиванию с типом Цел64. Оно определяет число элементов, выделенных для вектора, но не обязательно использованных. Если значение для выделить меньше или равно длине, то оно игнорируется.
- выражение для * должно быть *совместимым по присваиванию* с типом элемента вектора ($\S10.2$)
- *Индекс* должен быть выражением *совместимым по присваиванию* с типом Цел64
- Индекс должен быть константным выражением (§5.13)
- выражение для *Индекса* должно быть *совместимым по присваиванию* с типом элемента вектора (§10.2)
- если длина вектора явно задана, то значение всех индексов должно быть в диапазоне [0..длина-1]. Это ограничение проверяется во время компиляции, если длина задана константным выражением, иначе во время исполнения
- если в конструкторе задаются значения не для всех индексов, задание значения по умолчанию является обязательным
- повторное задание значения для элемента с одним индексом является ошибкой

Если длина конструируемого вектора не задана явно, то она равна максимальному индексу + 1.

Примеры конструкторов:

1

```
тип Числа = []Цел64
1
2
3
    пусть ч1 = Число[длина: 3, *: 0]
    // вектор [0, 0, 0]
4
5
6
    пусть ч2 = Число[длина: 5, *:0, 1: 1, 3: 3]
7
    // вектор [0, 1, 0, 3, 0]
8
    пусть ч3 = Число[*: 0, 1: 1, 3: 3]
9
10
    // [0, 1, 0, 3], длина - макс индекс + 1
```

Пример ошибки - значение по умолчанию не задано:

```
пусть ч3 = Число[1: 1, 3: 3]
```

Вектор является динамическим массивом, длина которого может изменяться во время выполнения. Конструктор вектора выделяет память под тах(длина, выделить) элементов. Если к вектору добавляются элементы (см. §7.2.1) и новое число элементов не превышает число выделенных, то выделения новой памяти не происходит.

```
1 пусть ч1 = Число[выделить: 100]
2 // вектор [], выделено место до 100 элементов
```

ТВО: методы выделено, выделить

5.8. Конструктор экземпляра класса

Конструктор вектора применяется к первичному выражению, которое должно обозначать тип класса. Результатом выполнения конструктора является экземпляр класса.

```
Конструктор-класса: '{' Список-значений? '}' Список-значений: Значение-поля (',' Значение-поля)* ','? Значение-поля: Идентификатор ':' Значение
```

Для каждого Значения поля идентификатор должен обозначать поле конструируемого класса. В результирующем экземпляре значением этого поля будет значение выражения Значение. Значение должно быть выражением, совместимым по присваиванию с типом поля (§10.2).

Указание нескольких значений для одного идентификатора поля является ошибкой.

Конструктор должен задавать значения для каждого поля класса с *поздней инициализацией* (§4.9.1). Если класс импортирован из другого модуля, в конструкторе не могут использоваться идентификаторы не экспортированных полей. Если при этом у не экспортированного поля задана *поздняя инициализация*, конструктор такого класса может быть использован только в модуле, содержащем описание класса.

```
тип Человек = класс {
    имя: Строка = позже
    возраст := 0

4 }

пусть Вася = Человек{имя: "Вася", возраст: 25}
```

Пример ошибки - значение поля "имя"с поздней инициализацией не задано:

```
1 пусть Некто := Человек{возраст: 25}
```

5.9. Подтверждение типа

Подтверждение типа применяется к первичному выражению, которое должно быть может быть типа ($\S4.10$).

```
Подтверждение-типа: '^'
```

Если значение первичного выражение, тип которого есть мб T равно 'пусто', то происходит *авария* (§6.13), иначе тип выражения становится равным T.

```
1 пусть кличка: мб Строка = пусто
2 ...
3 кличка = "Мурка"
4 ...
5 напечатать (кличка^)
```

5.10. Унарные операции

Унарная-операция:	- ' - ' - · ' ·	' ':~'

Операция	Действие	Разрешенные типы
_	отрицание	Байт, Цел64, Слово64, Вещ64
~	логическое НЕ	Лог
:~	инвертирование битов	Байт, Цел64, Слово64

Унарные операторы имеют приоритет выше, чем бинарные, то есть выражение -x+1 выполняется как (-x)+1. Тип результата унарных операций равен типу операнда.

5.11. Проверка типа

Операция *проверка типа* применима к операнду типа класс или *мб* класс. Тип, указанный в операции справа, должен быть типом класса. Тип результата - логический.

Для операнда мб типа со значением 'пусто' операция возвращает ложь. Иначе, операция проверяет, что динамический тип объекта равен типу, указанному в операции, или является расширением этого типа и возвращает значение истина, если это так.

```
Проверка-типа: Выражение 'типа' Указ-типа
```

```
тип К1 = класс {}
1
2
   тип К2 = класс (К1) {}
3
4
   пусть к1: К1 = К1{}
   пусть \kappa2: K1 = K2{}
5
6
   пусть к3: мб K1 = пусто
   пусть б := к1 типа К2 // ложь
8
   б := к2 типа К2 // истина
9
   б := к3 типа К2 // ложь
```

5.12. Бинарные операции

```
Бинарная-операция

: Арифметические
| Сравнения
| Логические
| Битовые
| Сдвиги
Арифметические: '+' | '-' | '*' | '/' | '%'
Сравнения: '=' | '#' | '<' | '<=' | '>' | '>='
Логические: '|' | '&'
Битовые: ':|' | ':&' | ':\'
Сдвиги: '<<' | '>>'
```

5.12.1. Приоритеты операции

Язык определяется пять уровней приоритета для бинарных операций. Операции умножения имеют самый высокий приоритет, а операция логическое ИЛИ (|) имеет самый низкий приоритет.

Приоритет	Операция
5	* / %
4	+ -
3	= # < <= > >=
2	&
1	

Бинарные операции с одинаковым приоритетом ассоциируются слева направо. Например, х / у * z эквивалентно (х / у) * z.

5.12.2. Арифметические операции

Арифметические операторы применяются к числовым операндам, выдают результат того же типа, что и тип операндов. Типы левого и правого операнда должны совпадать.

Операция	Действие	Разрешенные типы
+	сумма	Байт, Цел64, Слово64, Вещ64
-	разница	Байт, Цел64, Слово64, Вещ64
*	произведение	Байт, Цел64, Слово64, Вещ64
/	деление	Байт, Цел64, Слово64, Вещ64
%	остаток от деления	Байт, Цел64, Слово64

5.12.3. Операции сравнения

Операции сравнения выдают результат типа Лог.

Все операции сравнения '=', '#', '<', '<=', '>', '>=' применимы к операндам типов Байт, Цел64, Слово64, Цел64 и Символ. Типы левого и правого операнда, при этом, должны совпадать.

Операции '=', '#' также применимы в следующих случаях:

Тип левого операнда	Правый операнд
Лог	операнд типа Лог
Строка	операнд типа Строка
класс T	операнд типа T
мб T	'пусто' или операнд типа мб T

5.12.4. Логические операции

Операции логическое ИЛИ (|) и логическое И (&) применимы к операндам типа Лог и выдают результат типа Лог.

а | б означает если а, то истина, иначе б а & б означает если а, то б, иначе ложь

5.12.5. Битовые операции

Операции битовое ИЛИ (:|), битовое U(:&) и XOR - битовое исключающее ИЛИ $(:\:\setminus)$ применимы к операндам целых типов (Байт, Цел64, Слово64), оба операнда должны быть одного типа. Тип результата равен типу операндов.

5.12.6. Операции сдвига

Операции cdвиг влево (<<) и cdвиг вправо (>>) применимы к операндам целых типов и выдают результат того же типа, что и тип первого операнда.

5.13. Константные выражения

Константные выражения - это выражения, которые могут быть вычислены во время компиляции.

Минимальные требования к компилятору языка Тривиль: компилятор должен вычислять выражение, состоящее из одного операнда, который является литералом или идентификатором, обозначающим константу или функцию (не метод).

6. Операторы

Операторы задают действия.

Оператор

- : Локальное-описание
- | Простой-оператор
- | Оператор-если
- | Оператор-надо
- | Оператор-выбор
- | Оператор-пока
- | Оператор-цикл
- | Оператор-вернуть
- | Оператор-прервать
- | Оператор-авария

Простой-оператор

- : Оператор-выражение
- | Оператор-присваивания
- | Инкремент
- | Декремент

6.1. Блоки

Операторы сгрупированы в Блоки, которые задают последовательность действий для функций и методов (§4.11, §4.12) и входа в модуль (§8.3). Блок - это, возможно, пустая последовательность операторов, которые могут включать локальные описания.

```
Блок: '{' Список-операторов? '}'
Список-операторов: Оператор (Разделитель Оператор)*
```

6.2. Локальные описания

Локальное описание определяет идентификатор, для которого областью действия (§4.1) является часть блока, от точки завершения описания до завершающей скобки блока, исключая вложенные блоки, в которых описан такой же идентификатор.

```
Локальное-описание: Описание-переменной
```

Для локальных переменных не может быть задана *поздняя инициализация* (§4.6). **ТВD**: Нужны ли локальные константы? Или это задача оптимизации?

6.3. Выражение, как оператор

Выражение, являющееся вызовом функции или метода, кроме вызова стандартных функций ($\S7.1$), может быть использовано как оператор. Результат вызова при этом игнорируется.

Оператор-выражение: Выражение

```
1 напечатать ("Привет")
2 Факториал (5)
```

Не могут быть использованы в качестве оператора:

```
1 1 + 1 длина(a)
```

6.4. Оператор присваивания

Присваивание заменяет текущее значение, хранящееся в переменной, новым значением, заданным выражением.

```
Оператор-присваивания: Выражение ':=' Выражение
```

Выражение в левой части должно быть изменяемым:

Левое выражение		Условие
Имя объекта	пер или мод.пер	пер - это изменяемая перемен-
		ная (§4.6)
Доступ к полю	что-то.поле	что-то - это изменяемое выра-
		жение, поле - это изменяемое
		поле (§4.9.1)
Индексация	что-то[индекс]	что-то - это изменяемое выра-
		жение
Преобразование	что-то(:Тип)	что-то - это изменяемое выра-
		жение

Выражение в правой части должно быть *совместимо по присваиванию* с типом левого выражение ($\S10.2$).

```
1 пусть ц := 0
2 пусть байты := Байты[1, 2, 3]
3 ц := 2
4 байты[ц] = 5
```

Ошибка присваивания в неизменяемое выражение:

```
1 пусть ц = 0
2 ц := 2
```

6.5. Инкремент и декремент

Операторы *Инкремент* ('++') и *Декремент* ('--') увеличивают или уменьшают свои операнды на 1.

```
Инкремент: Выражение '++'
Декремент: Выражение '--'
```

Как и в случае присваивания (§6.4), выражение должно быть *изменяемым*. Тип выражения должен быть одного из типов: Байт, Цел64, Слово64.

```
1 пусть ц := 1
2 пусть байты := Байты[1, 2, 3]
3 байты[ц]++
```

6.6. Оператор 'если'

Оператор если определяет условное выполнение двух ветвей в соответствии со значением логического выражения. Если выражение принимает значение истина, выполняется Блок первой веткви, в противном случае, выполняется ветвь иначе, если она задана.

```
Оператор-если: 'если' Выражение Блок ('иначе' (Оператор-если | Блок))?
```

```
1 если ц > макс { ц := макс }
2 если Буква?(сим) { Имя() }
4 иначе если Цифра?(сим) { Число() }
5 иначе { Ошибка!() }
```

6.7. Оператор 'надо'

Оператор надо используется для завершения исполнения операторов некоторого контекста, если условие, заданое логическим выражением, не выполнено.

```
Оператор-надо:
    'надо' Выражение 'иначе' (Завершающий-оператор | Блок)
Завершающий-оператор
    : Оператор-вернуть
    | Оператор-прервать
    | Оператор-авария
```

Если в ветви иначе стоит $\mathit{Eлок}$, то последним оператором блока должен быть завершающий оператор.

Завершающий оператор	Действие
вернуть (§6.12)	выход из тела функции, метода или входа
прервать (§6.11)	выход из ближайшего объемлющего цикла
авария (§6.13)	аварийное завершение программы

```
1 надо делитель # 0 иначе авария("деление на ноль")
2 надо число > 1 иначе вернуть 1
```

6.8. Оператор 'выбор'

Оператор выбор выбирает одну ветвь выполнения из нескольких вариантов.

```
Оператор-выбор
: Выбор-по-выражению
| Выбор-по-предикатам
| Выбор-по-типу
```

Есть три разновидности оператора выбора

- выбор по выражению (§6.8.1)
- выбор по предикатам (§6.8.2)
- выбор по типу (§6.8.3)

6.8.1. Выбор по выражению

В операторе выбор по выражению выбор идет по по значению выражения оператора (значение сравнивается со значениями выражений в вариантах).

При выполнении вычисляется *Выражение* оператора, затем выражения в *Вариантах* вычисляются слева направо и сверху вниз. Как только значение выражения *Варианта* стало равным значению выражения оператора, выполняется список операторов этого *Варианта*. Остальные варинты игнорируются. Если ни один из вариантов не выполнен, и есть ветка другое, выполняется список операторов этой ветки.

Тип каждого выражения в вариантах должен быть равен эквивалентен типу первого выражения (§10.1).

```
выбор x {
2 когда 0: вернуть "ничего"
3 когда 1: вернуть "один"
4 когда 2: вернуть "два"
5 другое вернуть "много"
6 }
```

6.8.2. Выбор по предикатам

Если выражение после ключевого слова выбор отсутствует, то это оператор *выбора по предикатам*.

В этом случае, каждое выражение в Вариантах должно быть выражением логического типа. Эти выражения вычисляются слева направо и сверху вниз до тех пор, когда значение выражения не равно истине. Как только значение выражения стало рано истине, выполняется список операторов этого варианта. Остальные варинты игнорируются. Если ни один из вариантов не выполнен, и есть ветка другое, выполняется список операторов этой ветки.

```
1 выбор x {
2 когда x < 0: вернуть "отрицательное"
3 когда x > 0: вернуть "положительное"
4 другое вернуть "ноль"
5 }
```

6.8.3. Выбор по типу

В операторе выбор по типу выбор ветви выполнения происходит по типу выражения.

```
Выбор-по-выражению:
    'выбор' Переменная-варианта? 'тип' Выражение '{'
    Вариант-типа*
    ('другое' Список-операторов?)?
    '}'
Переменная-варианта:
    'пусть' Идентификатор ':'
Вариант-типа:
```

```
'когда' Указ-типа (',' Указ-типа)* ':'
Список-операторов?
```

Рассмотрим сначала оператор, в котором переменная варианта не указана.

При выполнении такого оператора вычисляется Выражение, тип которого должен быть типом класса (§4.9). Для дальнейшего сравнения используется тип объекта, то есть тип, с которым объект был создан, иначе говоря, динамический тип.

В следующем примере для сравнения будет использовать тип К2, так как это тип объекта к:

```
тип K1 = класс {}

тип K2 = класс (K1) {}

вход{

пусть к: K = K2{}

выбор тип к { ... }

}
```

Динамический тип выражения сравнивается слева направо и сверху вниз на равенство типам, указанным в вариантах. Как только один из типов *Вариантаа* оказался равен типу выражения, выполняется список операторов этого *Вариантаа*. Остальные варинты игнорируются. Если ни один из вариантов не выполнен, и есть ветка другое, выполняется список операторов этой ветки.

Сравнение типа выражение оператора выполняется на строгое равентсво, наследование не учитывается. Таким образом, в примере сработает ветка когда К2, независимо от порядка веток в операторе.

```
тип К1 = класс {}
1
2
   тип К2 = класс (К1) {}
3
   вход{
4
       пусть к: K = K2\{\}
5
       выбор тип к {
       когда К1:
6
        когда К2: // сработает эта ветка
7
8
9
```

Если в операторе указана *переменная варианта*, то в каждой ветки исполнения (но не в ветке другое) будет доступна переменная с именем, заданным идентификатором, и типом, указанном в варианте. При этом в варианте должен быть указан только один тип.

```
тип Человек = класс { имя := "" }
1
    тип Работник = класс (Человек) {
2.
        зарплата := 0.0
3
4
    фн зарплата (чел: Человек): Вещ64 {
5
6
        выбор пусть х: тип чел {
7
        когда Работник: вернуть х.зарплата
8
        другое вернуть 0.0
9
        }
10
```

Переменная варианта является *переменной с единственным присваиванием*, её значение изменить нельзя.

6.9. Оператор 'пока'

Оператор пока определяет повторное выполнение блока до тех пор, пока значение логического выражение равно истина. Выражение вычисляется перед каждой итерацией.

```
Оператор-пока: 'пока' Выражение Блок
```

```
1 пока a < б {
2 a := a * 2
3 }
```

6.10. Оператор цикл

Оператор цикл определяет выполнение блока для всех элементов вектора или другого *индексируемого* объекта.

```
Оператор-цикл:
   'цикл' ПеременныеЦикла 'среди' Выражение Блок
ПеременныеЦикла
   : ПеременнаяЦикла
   | '[' ПеременнаяЦикла ']' ПеременнаяЦикла?
ПеременнаяЦикла: Идентификатор
```

Выражение цикла должно быть индексируемым, то есть

- тип выражения должен быть типом вектора
- или выражение должно обозначать вариативный параметр.

Оператор позволяет задать две переменные цикла или только одну из них. В теле цикла, значением *переменной индекса*, заданной в квадратных скобках, является текущий индекс (от 0 до длины-1), а *переменной элемента* - значение элемента по этому индексу.

Типы переменных цикла задаются неявно:

- тип переменной индекса есть Цел 64
- тип переменной элемента равен типу элемента индексируемого выражения

ТВД: Надо ли задавать типы переменных цикла явно?

```
пусть числа = Числа[9, 8, 7]
цикл [№] число среди числа {
вывод.ф("%v:%v, ", №, число)
}
// вывод: 0:9, 1:8, 2:7,
```

Каждая переменная цикла является *переменной с единственным присваиванием*, её значение в теле цикла изменить нельзя.

6.11. Оператор 'прервать'

Оператор прервать завершает выполнение самого внутреннего оператора цикла в рамках одной и той же функции.

```
Оператор-прервать: 'прервать'
```

```
1 пока истина {
2 а := а * 2
3 если а > б { прервать }
4 }
```

6.12. Оператор 'вернуть'

Оператор вернуть завершает выполнение тела функции, метода или входа, и, возможно, возращает значение.

```
Оператор-вернуть: 'вернуть' (Выражение | Разделитель)
```

Если оператор вернуть используется в теле входа или в теле функции или метода, в сигнатуре которых не указан тип результата, то выражение в операторе должно отсутствовать.

И наоборот, если в оператор используется в теле функции или метода, в сигнатуре которых указан тип результата, выражение должно быть и оно должно быть совместимо по присваиванию с типом результата (§10.2).

```
фн Факториал(ц: Цел64): Цел64 {
    если ц <= 1 { вернуть 1 }
    вернуть ц * Факториал(ц - 1)
}
```

Если выражение в операторе присутствует, то оно должно начинаться на одной строке с ключевым словом вернуть, иначе компилятор выдаст ошибку:

```
1 фн цифра?(сим: Символ): Лог {
2 вернуть // ошибка компиляции
3 '0' <= сим & сим <= '9'
4 }
```

А так верно:

```
1 фн цифра?(сим: Символ): Лог {
2 вернуть '0' <= сим
3 & сим <= '9'
4 }
```

6.13. Оператор 'авария' и аварийная ситуация

Оператор авария запускает аварийную ситуация, которая, как правило, приводит к аварийному завершению программы.

```
Оператор-авария: 'авария' '(' Выражение ')'
```

Тип выражения должен быть Строка. Значение этого выражения используется в сообщении об аварийной ситуации.

Кроме явного запуска аварийной ситуации, аварийная ситуация может быть запущена неявно, в следующих случаях:

- Выход индекса за границы вектора при индексации
- Операция подтверждения типа выполнена над объектом со значением 'пусто'
- Недопустимое преобразование типа
- Недопустимая последовательность байт в кодировке UTF-8
- Невозможность выделения памяти для динамического объекта
- Нереализованная возможность в системе поддержки выполнения

Язык Тривиль не содержит средств перехвата аварийной ситуации или восстановления после него. Такие средства могут быть добавлены на уровне библиотек.

7. Стандартные функции и методы

7.1. Стандартные функции

7.1.1. Длина

Стандартная функция длина возвращает длину строк, векторов и вариативных параметров, см. также (§4.3.2). Тип результата: Цел64.

Длина строки считается в символах. Для остальных типов длина выдает число элементов.

```
1 пусть д1 = длина("Привет") // равно 6
2
3 пусть д2 = длина("Привет"(:Байты)) // равно 12
```

7.1.2. Функции для полиморфных параметров

Стандартные функции тег и нечто предназначены для работы с полиморфными параметрами (§4.11.2)

Аргумент функция тег может быть типом или полиморфным параметром или элементом вариативного полиморфного параметра. Для каждого типа определен его *Тег*, который является постоянным объектом времени исполнения. Постоянный, в данном контексте, контексте, означает, что Тег типа не меняется во время исполнения программы. Тег представлен значением типа Слово64.

Вид аргумента	Пример	Результат
предопределенный тип	тег(Цел64)	Тег указанного типа
описанный тип	тег(Байты)	Тег указанного типа
полиморфный параметр	тег(п)	Тег параметра
элемент вариативного поли-	тег(п[номер])	Тег элемента
морфного параметра		

Аргумент функция нечто должен быть полиморфным параметром или элементом вариативного полиморфного параметра. Функция выдает результат типа Слово64, которые является скрытым представлением значения переданного в полиморфный параметр.

К результату функции нечто может применяется преобразование типа (§5.5) или *осторожное* преобразование типа (§11.1).

```
1 фн Что это?(п: *) {
2 если тег(п) = тег(Цел64) {
3 вывод.ф("это число: %v", нечто(п)(:Цел64))
4 }
5 иначе {
6 вывод.ф("это не число")
7 }
8 }
```

7.2. Встроенные методы для векторов

Для всех типов векторов определены стандартные методы:

- добавить
- заполнить

TBD: Уточнить семантику и набор методов.

7.2.1. Метод добавить

Метод добавить можно применить к любому вектору. Если вектор типа $[\]$ T, то метод можно рассматривать, как описанный с заголовком:

```
1 фн (в: []Т) добавить(аргументы: ...Т)
```

Это означает, что в вызове метода можно указать любое количество аргументов типа $\mathbb T$ или одно развернутое выражение ($\S 5.6$).

Вызов метода добавляет к вектору указанные аргументы.

```
тип Числа = []Цел64

пусть фиб = Числа[1, 1]
фиб.добавить (2, 3, 5)
```

8. Модули

Программа на языке Тривиль состоит из модулей (единиц компиляции), исходный текст каждого модуля расположен в одном или нескольких исходных файлов.

Каждый исходный файл состоит из заголовка модуля, за которым следует, возможно, пустой список импорта, за которым следует, возможно, пустой набор описаний типов, констант, переменных, функций и методов и, возможно, вход в модуль.

```
Модуль: Исходный-файл+
Исходный-файл:
Заголовок-модуля
Список-импорта
Описание-или-вход*

Заголовок-модуля:
'модуль' Идентификатор Разделитель
('осторожно' Разделитель)?

Описание-или-вход: Описание | Вход
```

8.1. Заголовок модуля

Каждый исходный файл начинается с заголовка модуля. Заголовок содержит идентификтор, который определяет модуль, к которому принадлежит файл. Идентификатор не принадлежит никакой области действия и не связан ни с каким объектом.

Заголовок модуля может содержать признак осторожно, который означает, что в исходном файле можно использовать *небезопасные операции* (§11).

8.2. Импорт

Наличие импорта в исходном файле указывает, что этот исходный файл зависит от функциональности импортируемого модуля. Импорт обеспечивает доступ к экспортированным идентификаторам импортируемого модуля.

Каждый импорт содержит путь импорта, указывающий на размещение импортируемого модуля.

```
Список-импорта: Импорт*
Импорт: 'импорт' Путь-импорта Разделитель
Путь-импорта: '"' Путь-в-хранилище | Файловый-путь '"'
```

TBD: явное задание имени импорта.

Импорт добавляет описание идентификатора, текстуально равного последнему имени в пути импорта, который будет использоваться для доступа к экспортированным объектам импортируемого модуля.

```
1 модуль х

2 импорт "стд::вывод"

4 вход {

вывод.ф("Привет!")

7 }
```

Путь импорта может быть задан как *путь в хранилище* исходных текстов или как *файловый путь*. В любом случае, в итоге, он должен указывать на папку, содержащую исходные файлы импортируемого модуля.

```
Путь-в-хранилище:
Имя-хранилища '::' Имя-папки ('/' Имя-папки)*
```

Путь в хранилище состоит из имени хранилища, которое явным или неявным образом (см. README) должно быть привязано к определенной папки файловой системы и пути, относительно этой папки:

```
1 импорт "стд::юникод/utf8"
2 пусть декодер = utf8.декодер(...)
```

Язык не определяет формально набор символов, которые могут использоваться в *имени хранилища* и *имени папки*, надеясь на здравый смысл разработчиков.

Файловый путь может быть аболютным или относительным (в терминах инструментальной платформы). Не рекомендуется использовать абсолютные пути, так как это приводит к непереносимому коду. Относительный путь всегда трактуется как путь относительно той папки, в которой запущен компилятор (рабочая папка).

Например в случае импорта:

```
импорт "трик/лексер"
```

будет импортирован модуль, который расположен в подпапке 'трик/лексер' рабочей папки.

Имена папок должны разделяться символом / (слеш) независимо от инструментальной платформы, на которой используется компилятор.

8.3. Вход или инициализация модуля

Модуль может содержать действия, которые выполняются при инициализации модуля - *вход в модуль*.

```
1 Вход: 'вход' Блок
```

TBD: описать: переменные с поздней инициализацией

Система должна обеспечивать следующие условия инициализации модуля М:

• инициализация модуля выполняется один раз

- инициализации M выполняется **после** инициализации всех модулей, которые импортирует M
- инициализации M выполняется **до** инициализации тех модулей, которые импортируют M

8.4. Инициализация и исполнение программы

Программа состоит из головного модуля и всех прямо или косвенно импортированных модулей.

Исполнение программы состоит из:

- Инициализации всех модулей, импортированных из головного модуля. Это приводит к рекурсивной инициализации всех используемых модулей программы. Для корректной инициализации граф импорта должен быть ациклическим.
- И, затем, выполнение входа головного модуля.

9. Обобщенные модули

Практически все современные языки программирования содержат конструкции для определения обобщенных (generic) типов и функций. Одна из важнейших областей применения обобщенных конструкций - библиотечная реализация полиморфных контейнеров: стек, очередь, деревья, словари (map).

Как правило, семантика обобщенных типов и их реализации весьма нетривиальна, и тем самым не подходит для Тривиль.

Для языка Тривиль выбрано решение предельно простое во всех отношениях. Обобщение в нем сделано на уровне модулей, то есть только модуль целиком может быть обобщенным или, другими словами, параметризованным. Причем, он может быть параметризован не только типом, а также любым языковым объектом, например, константой или функцией.

Обобщенный модуль - это модуль, в котором опущено то определение (или определения), которым он параметризован. Другими словами - обобщенный модуль является недопределенным.

Рассмотрим, в качестве примера, обобщенный модуль Стек (полный текст см. "стд/контейнеры/стек"):

```
1
    модуль стек
2
3
    тип Элементы = []Элемент
4
5
    тип Стек* = класс {
         элементы = Элементы[]
6
7
        верх: Цел64 := 0 // ... [0..верх[
8
9
    фн (с: Стек) положить*(э: Элемент) {
10
11
        если с.верх = длина (с.элементы) {
12
             с.элементы.добавить (э)
13
         } иначе {
             с.элементы[с.верх] := э
14
15
         с.верх++
16
17
       другие методы */
18
```

Тип элемента стека - это Элемент, и этот тип не определен в самом модуле.

Для использования конкретного стека, например, стека целых чисел, необходимо определить *настройку*. Настройка - отдельный модуль, в котором указан путь к обобщенному модулю и заданы параметры обобщенного модуля:

```
1 настройка "стд/контейнеры/стек"
2 модуль стек-цел
3 тип Элемент = Цел64
```

Настроенный модуль используется обычным образом. Например, если стекцел размещен в папке "модули/стек-цел", то использование его выглядит так.

1 модуль пример

2 импорт "модули/стек-цел"

4 вход {
 пусть с = стек-цел.Стек{}
 с.положить(1)

8 }

Несколько настроек можно использовать одновременно, например, если модель стек настроен также на Строку:

```
1
    модуль пример
2
3
    импорт "стд/вывод"
    импорт "модули/стек-цел"
4
    импорт "модули/стек-строк"
5
6
7
    вход {
8
        пусть c1 = cтek-цел.Cтek{}
9
        пусть c2 = cтек-строк.Стек{}
10
11
         с1. положить (777)
         с2. положить ("привет")
12
```

9.1. Последствия принятого решения

Сравнение подхода с другими известными подходами выходит за рамки описания языка. Замечу только, что главной причиной выбора такого подхода является тривиальность реализации и простота использования.

Прямым (и несколько непривычным) следствием подхода является то, что обобщенный модуль не может быть использован без настройки. Попытка импорта обобщенного модуля приведет к ошибкам компиляции.

Перед компиляцией компилятор соберет модуль, например, стек-цел, из двух частей:

- из обобщенного модуля, в котором параметры не определены
- и из модуля с настройкой, в котором параметры определены

Такой собранный модуль является полностью определенным, он компилируется обычным образом. Каждый настроенный модуль компилируется отдельно и для него строится отдельный код. Так код для стек-цел никак не связан с кодом для стек-строк.

Говоря в принятых терминах, Тривиль использует мономорфизацию на уровне исходного текста, тем самым получая оптимальный по скорости код за счет увеличения размера кода. Вся работа с обобщенными модулями происходит во время компиляции, во время исполнения все модули являются полностью определенными.

Замечу, что никаких синтаксических конструкций для ограничения параметров (constraints) в языке не предусмотрено, они накладываются неявно. Например,

если в обобщенном модуле использовать операцию <, то настройка на арифметический тип будет работать, а настройка на тип, для которого эта операция не определена, приведет к ошибкам компиляции.

9.2. Уточненный синтаксис модуля

Уточненный синтаксис модуля с добавлением опциональной настройки (исходное определение см. §8):

```
Модуль: Исходный-файл+
Исходный-файл:
    ('настройка' Путь-импорта)?
    Заголовок-модуля
    Список-импорта
    Описание-или-вход*
```

10. Правила совместимости

10.1. Эквивалентность типов

Во многих случаях, например, для большинства бинарных операция или при переопределения методов, требуется эквивалентность типов.

Два типа *T1* и *T2* эквивалентны, если выполняется одно из условий:

- типы *T1* и *T2* это один и тот же тип, возможно, переименованный, так как переименование типа не создает новый тип
- типы T1 и T2 это типы векторов и типы их элементов эквивалентны
- типы Т1 и Т2 это может быть типы и их базовые типы эквивалентны

Переименование типа не создает новый тип, таким образом, после описания:

```
тип Числа = Цел64
пусть ч: Число = 1
пусть ц: Цел64 = 2
пусть с: Слово64 = 3
```

10.2. Совместимость по присваиванию

Совместимость по присваиванию используется в следующих случаях:

- инициализация констант и переменных с явным типом: пусть x: T = выражение, выражение должно быть совместимо с типом T
- оператор присваивания: х := выражение
- оператор вернуть выражение, выражение должно быть совместимо с типом результата функции
- передача фактических параметров при вызове: Φ (аргумент1, ..., аргументN), аргумент должен быть совместим с типом параметра

Значение типа TB совместимо по присваиванию с целевым типом $T\Pi$ (типом переменной, константы или параметра), если выполняется одно из условий:

- ТП и ТВ эквивалентны
- $T\Pi$ это целый тип (Байт, Цел64, Слово64), а значение это литерал целого типа, причем значение литерала входит в допустимый диапазон значений $T\Pi$
- $T\Pi$ и TB это типы векторов и типы их элементов эквивалентны
- $T\Pi$ это $m\delta$ T1, а значение есть 'пусто' или $T\Pi$ это $m\delta$ T2 и типы T1 и T2 эквивалентны
- $T\Pi$ и TB это типы классов и TB является расширением $T\Pi$
- переменная является полиморфным параметром

11. Применять с осторожностью

Низкоуровневые типы и операции.

11.1. Осторожные преобразования типа

Часть преобразований типа являются низкоуровневыми и/или требуют особого внимания разработчика. Для использования таких преобразований операция преобразования должна быть помечана ключевым словом осторожно, кроме того исходный файл, в котором используется *небезопасное преобразование* тоже должен быть помечен ключевым словом осторожно (§8).

```
Преобразование: '(:' 'осторожно'? Указ-типа ')'
```

Разрешенные преобразования:

Целевой тип	Тип выражения
Цел64	Слово64
Вещ64	Слово64
Слово64	Цел64
Слово64	Вещ64
Слово64	ссылочный тип
ссылочный тип	Слово64

Ссылочными типами являются Строка, типы вектора и типы класса.

Все преобразования выполняются без изменения битового представления. Для преобразования Слово64 в ссылочный тип выполняется проверка времени исполнения, и запускается *аварийная ситуация* (§6.13) в случае несовпадения целевого типа и динамического типа выражения.

А. Версии языка

А.1. Версия 0.9.0 от 07.09.2023

• Добавлен тип Пусто с единственным значением - литералом пусто (§4.3)

А.2. Версия 0.9.1 от 07.09.2023

• Разрешен возврат из функции полиморфного типа (§4.11)